

INTRODUCTION TO SCHEME

COMPUTER SCIENCE MENTORS CS 61A

April 1 to April 3, 2019

Scheme is a programming language, much like Python. In fact, many of Python's design features were inspired by Scheme. The point of learning this language is twofold: one, we're looking into what parts of Python generalize to other languages. Two, we want to start thinking about how to design and build (an interpreter for) a programming language, and it turns out Scheme is a nice one to build. In fact, we'll show you enough of the language in this hour to write recursive procedures. This section covers the basics. You'll learn the rest in lab and discussion. It's pretty awesome that we'll be picking up a whole new programming language within an hour.

Visit scheme.cs61a.org to try the online interpreter. Type `(autodraw)` and the interpreter will automatically draw box-and-pointer diagrams whenever an expression evaluates to a Scheme pair.

1 What Would Scheme Print?

1. What will Scheme output?

```
scm> (define pi 3.14)
```

Solution: pi

```
scm> pi
```

Solution: 3.14

```
scm> 'pi
```

Solution: pi

```
scm> (+ 1 2)
```

Solution: 3

```
scm> (+ 1 (* 3 4))
```

Solution: 13

```
scm> (if 2 3 4)
```

Solution: 3

```
scm> (if 0 3 4)
```

Solution: 3

```
scm> (- 5 (if #f 3 4))
```

Solution: 1

```
scm> (if (= 1 1) 'hello 'goodbye)
```

Solution: hello

```
scm> (define (factorial n)
      (if (= n 0)
          1
          (* n (factorial (- n 1)))))
```

Solution: factorial

```
scm> (factorial 5)
```

Solution: 120

2 Code Writing in Scheme

2. **Hailstone yet again** Define a program called `hailstone`, which takes in two numbers `seed` and `n`, and returns the n th hailstone number in the sequence starting at `seed`. Assume the hailstone sequence starting at `seed` is longer or equal to `n`. As a reminder, to get the next number in the sequence, if the number is even, divide by two. Else, multiply by 3 and add 1.

Useful procedures

- `quotient`: floor divides, much like `//` in python
`(quotient 103 10)` outputs 10
- `remainder`: takes two numbers and computes the remainder of dividing the first number by the second
`(remainder 103 10)` outputs 3

```
; The hailstone sequence starting at seed = 10 would be  
; 10 => 5 => 16 => 8 => 4 => 2 => 1
```

```
; Doctests  
> (hailstone 10 0)  
10  
> (hailstone 10 1)  
5  
> (hailstone 10 2)  
16  
> (hailstone 5 1)  
16
```

```
(define (hailstone seed n)
```


```
)
```

Solution:

```
(define (hailstone seed n)
  (if (= n 0)
      seed
      (if (= 0 (remainder seed 2))
          (hailstone
            (quotient seed 2)
            (- n 1))
          (hailstone
            (+ 1 (* seed 3))
            (- n 1)))))
```

3 Scheme Lists

```
scm> (cons 1 (cons 2 nil))
```


Solution: (1 2)

```
scm> (cons 1 '(2 3 4 5))
```

Solution: (1 2 3 4 5)

```
scm> (cons 1 '(2 (cons 3 nil)))
```

Solution: (1 2 (cons 3 ()))



```
scm> (cons 1 (2 (cons 3 nil)))
```

Solution:

eval: bad function **in** : (2 (cons 3 nil))

```
scm> (cons 3 (cons (cons 4 nil) nil))
```

Solution: (3 (4))

```
scm> (define a '(1 2 3))
```

Solution:

a

```
scm> a
```

Solution:

(1 2 3)

```
scm> (car a)
```

Solution:

1

```
scm> (cdr a)
```

Solution:

(2 3)

```
scm> (cadr a)
```

Solution:

2

How can we get the 3 out of a?

Solution:

```
(car (cdr (cdr a)))
```

4 More Code Writing in Scheme

3. Define `waldo` which takes in a list. If that list contains the symbol `waldo`, it returns the index where `waldo` first appears. Otherwise, it returns `#f`.

```
scm> (waldo '(1 4 waldo))
```

```
2
```

```
scm> (waldo '())
```

```
#f
```

```
scm> (waldo '(1 4 9))
```

```
#f
```

```
(define (waldo lst)
```

```
)
```

Solution:

```
(define (waldo lst)
  (define (helper lst index)
    (cond ((null? lst) #f)
          ((eq? (car lst) 'waldo) index)
          (else (helper (cdr lst) (+ index 1)))))
  (helper lst 0))
```